Index: configure.ac
===================================================================
--- configure.ac	(révision 956)
+++ configure.ac	(copie de travail)
@@ -18,8 +18,9 @@
 AC_C_INLINE
 
 # Checks for programs.
+AC_LANG([C++])
 AC_PROG_CXX
-#AC_PROG_CC
+AX_CXX_COMPILE_STDCXX_11([noext], [mandatory])
 
 # Checks for library functions.
 AC_FUNC_SELECT_ARGTYPES
@@ -253,7 +254,7 @@
 
 if test "$enable_debug" = "yes"; then
   CXXFLAGS="$CXXFLAGS -DDEBUG -g"
-  LDFLAGS="$LDFLAGS -g -ansi -Wall"
+  LDFLAGS="$LDFLAGS -g -Wall"
 fi
 
 # =========================================================================
@@ -265,13 +266,13 @@
 
 if test "$enable_profile" = "yes"; then
   CXXFLAGS="$CXXFLAGS -pg"
-  LDFLAGS="$LDFLAGS -pg -ansi -Wall"
+  LDFLAGS="$LDFLAGS -pg -Wall"
 fi
 
 # =========================================================================
 # gcc flags
 
-CXXFLAGS="$CXXFLAGS -ansi -Wall -Wextra -Wconversion" #-Werror
+CXXFLAGS="$CXXFLAGS -Wall -Wextra -Wconversion" #-Werror
 
 # =========================================================================
 # make use of ccache
Index: m4/ax_cxx_compile_stdcxx_11.m4
===================================================================
--- m4/ax_cxx_compile_stdcxx_11.m4	(nonexistent)
+++ m4/ax_cxx_compile_stdcxx_11.m4	(copie de travail)
@@ -0,0 +1,172 @@
+# ============================================================================
+#  http://www.gnu.org/software/autoconf-archive/ax_cxx_compile_stdcxx_11.html
+# ============================================================================
+#
+# SYNOPSIS
+#
+#   AX_CXX_COMPILE_STDCXX_11([ext|noext],[mandatory|optional])
+#
+# DESCRIPTION
+#
+#   Check for baseline language coverage in the compiler for the C++11
+#   standard; if necessary, add switches to CXXFLAGS to enable support.
+#
+#   The first argument, if specified, indicates whether you insist on an
+#   extended mode (e.g. -std=gnu++11) or a strict conformance mode (e.g.
+#   -std=c++11).  If neither is specified, you get whatever works, with
+#   preference for an extended mode.
+#
+#   The second argument, if specified 'mandatory' or if left unspecified,
+#   indicates that baseline C++11 support is required and that the macro
+#   should error out if no mode with that support is found.  If specified
+#   'optional', then configuration proceeds regardless, after defining
+#   HAVE_CXX11 if and only if a supporting mode is found.
+#
+# LICENSE
+#
+#   Copyright (c) 2008 Benjamin Kosnik <bkoz@redhat.com>
+#   Copyright (c) 2012 Zack Weinberg <zackw@panix.com>
+#   Copyright (c) 2013 Roy Stogner <roystgnr@ices.utexas.edu>
+#   Copyright (c) 2014, 2015 Google Inc.; contributed by Alexey Sokolov <sokolov@google.com>
+#   Copyright (c) 2015 Paul Norman <penorman@mac.com>
+#
+#   Copying and distribution of this file, with or without modification, are
+#   permitted in any medium without royalty provided the copyright notice
+#   and this notice are preserved. This file is offered as-is, without any
+#   warranty.
+
+#serial 13
+
+m4_define([_AX_CXX_COMPILE_STDCXX_11_testbody], [[
+  template <typename T>
+    struct check
+    {
+      static_assert(sizeof(int) <= sizeof(T), "not big enough");
+    };
+
+    struct Base {
+    virtual void f() {}
+    };
+    struct Child : public Base {
+    virtual void f() override {}
+    };
+
+    typedef check<check<bool>> right_angle_brackets;
+
+    int a;
+    decltype(a) b;
+
+    typedef check<int> check_type;
+    check_type c;
+    check_type&& cr = static_cast<check_type&&>(c);
+
+    auto d = a;
+    auto l = [](){};
+    // Prevent Clang error: unused variable 'l' [-Werror,-Wunused-variable]
+    struct use_l { use_l() { l(); } };
+
+    // http://stackoverflow.com/questions/13728184/template-aliases-and-sfinae
+    // Clang 3.1 fails with headers of libstd++ 4.8.3 when using std::function because of this
+    namespace test_template_alias_sfinae {
+        struct foo {};
+
+        template<typename T>
+        using member = typename T::member_type;
+
+        template<typename T>
+        void func(...) {}
+
+        template<typename T>
+        void func(member<T>*) {}
+
+        void test();
+
+        void test() {
+            func<foo>(0);
+        }
+    }
+
+    // Check for C++11 attribute support
+    void noret [[noreturn]] () { throw 0; }
+]])
+
+AC_DEFUN([AX_CXX_COMPILE_STDCXX_11], [dnl
+  m4_if([$1], [], [],
+        [$1], [ext], [],
+        [$1], [noext], [],
+        [m4_fatal([invalid argument `$1' to AX_CXX_COMPILE_STDCXX_11])])dnl
+  m4_if([$2], [], [ax_cxx_compile_cxx11_required=true],
+        [$2], [mandatory], [ax_cxx_compile_cxx11_required=true],
+        [$2], [optional], [ax_cxx_compile_cxx11_required=false],
+        [m4_fatal([invalid second argument `$2' to AX_CXX_COMPILE_STDCXX_11])])
+  AC_LANG_PUSH([C++])dnl
+  ac_success=no
+  AC_CACHE_CHECK(whether $CXX supports C++11 features by default,
+  ax_cv_cxx_compile_cxx11,
+  [AC_COMPILE_IFELSE([AC_LANG_SOURCE([_AX_CXX_COMPILE_STDCXX_11_testbody])],
+    [ax_cv_cxx_compile_cxx11=yes],
+    [ax_cv_cxx_compile_cxx11=no])])
+  if test x$ax_cv_cxx_compile_cxx11 = xyes; then
+    ac_success=yes
+  fi
+
+  m4_if([$1], [noext], [], [dnl
+  if test x$ac_success = xno; then
+    for switch in -std=gnu++11 -std=gnu++0x; do
+      cachevar=AS_TR_SH([ax_cv_cxx_compile_cxx11_$switch])
+      AC_CACHE_CHECK(whether $CXX supports C++11 features with $switch,
+                     $cachevar,
+        [ac_save_CXXFLAGS="$CXXFLAGS"
+         CXXFLAGS="$CXXFLAGS $switch"
+         AC_COMPILE_IFELSE([AC_LANG_SOURCE([_AX_CXX_COMPILE_STDCXX_11_testbody])],
+          [eval $cachevar=yes],
+          [eval $cachevar=no])
+         CXXFLAGS="$ac_save_CXXFLAGS"])
+      if eval test x\$$cachevar = xyes; then
+        CXXFLAGS="$CXXFLAGS $switch"
+        ac_success=yes
+        break
+      fi
+    done
+  fi])
+
+  m4_if([$1], [ext], [], [dnl
+  if test x$ac_success = xno; then
+    dnl HP's aCC needs +std=c++11 according to:
+    dnl http://h21007.www2.hp.com/portal/download/files/unprot/aCxx/PDF_Release_Notes/769149-001.pdf
+    dnl Cray's crayCC needs "-h std=c++11"
+    for switch in -std=c++11 -std=c++0x +std=c++11 "-h std=c++11"; do
+      cachevar=AS_TR_SH([ax_cv_cxx_compile_cxx11_$switch])
+      AC_CACHE_CHECK(whether $CXX supports C++11 features with $switch,
+                     $cachevar,
+        [ac_save_CXXFLAGS="$CXXFLAGS"
+         CXXFLAGS="$CXXFLAGS $switch"
+         AC_COMPILE_IFELSE([AC_LANG_SOURCE([_AX_CXX_COMPILE_STDCXX_11_testbody])],
+          [eval $cachevar=yes],
+          [eval $cachevar=no])
+         CXXFLAGS="$ac_save_CXXFLAGS"])
+      if eval test x\$$cachevar = xyes; then
+        CXXFLAGS="$CXXFLAGS $switch"
+        ac_success=yes
+        break
+      fi
+    done
+  fi])
+  AC_LANG_POP([C++])
+  if test x$ax_cxx_compile_cxx11_required = xtrue; then
+    if test x$ac_success = xno; then
+      AC_MSG_ERROR([*** A compiler with support for C++11 language features is required.])
+    fi
+  else
+    if test x$ac_success = xno; then
+      HAVE_CXX11=0
+      AC_MSG_NOTICE([No compiler with C++11 support was found])
+    else
+      HAVE_CXX11=1
+      AC_DEFINE(HAVE_CXX11,1,
+                [define if the compiler supports basic C++11 syntax])
+    fi
+
+    AC_SUBST(HAVE_CXX11)
+  fi
+])
Index: plugins/actions/dialoguize/dialoguize.cc
===================================================================
--- plugins/actions/dialoguize/dialoguize.cc	(révision 956)
+++ plugins/actions/dialoguize/dialoguize.cc	(copie de travail)
@@ -23,7 +23,7 @@
  *	along with this program. If not, see <http://www.gnu.org/licenses/>.
  */
  
-#include <auto_ptr.h>
+#include <memory.h>
 #include "extension/action.h"
 #include "i18n.h"
 #include "debug.h"
Index: plugins/actions/documentmanagement/documentmanagement.cc
===================================================================
--- plugins/actions/documentmanagement/documentmanagement.cc	(révision 956)
+++ plugins/actions/documentmanagement/documentmanagement.cc	(copie de travail)
@@ -178,9 +178,9 @@
 
 		ui_id = ui->new_merge_id();
 
-		#define ADD_UI(name) ui->add_ui(ui_id, "/menubar/menu-file/"name, name, name);
-		#define ADD_OPEN_UI(name) ui->add_ui(ui_id, "/menubar/menu-file/menu-open/"name, name, name);
-		#define ADD_SAVE_UI(name) ui->add_ui(ui_id, "/menubar/menu-file/menu-save/"name, name, name);
+		#define ADD_UI(name) ui->add_ui(ui_id, Glib::ustring::compose("%1/%2","/menubar/menu-file", name), name, name);
+		#define ADD_OPEN_UI(name) ui->add_ui(ui_id, Glib::ustring::compose("%1/%2","/menubar/menu-file/menu-open", name), name, name);
+		#define ADD_SAVE_UI(name) ui->add_ui(ui_id, Glib::ustring::compose("%1/%2","/menubar/menu-file/menu-save", name), name, name);
 
 		ADD_UI("new-document");
 		ADD_OPEN_UI("open-document");
Index: src/subtitleview.cc
===================================================================
--- src/subtitleview.cc	(révision 956)
+++ src/subtitleview.cc	(copie de travail)
@@ -1363,7 +1363,7 @@
 	{
 		int num;
 		std::istringstream ss(event->string);
-		bool is_num = ss >> num != 0; 
+		bool is_num = static_cast<bool>(ss >> num) != 0; 
 		// Update only if it's different
 		if(is_num != get_enable_search())
 			set_enable_search(is_num);
Index: src/utility.h
===================================================================
--- src/utility.h	(révision 956)
+++ src/utility.h	(copie de travail)
@@ -91,7 +91,7 @@
 	std::istringstream s(src);
 	// return s >> dest != 0;
 
-	bool state = s >> dest != 0;
+	bool state = static_cast<bool>(s >> dest) != 0;
 
 	if(!state)
 		se_debug_message(SE_DEBUG_UTILITY, "string:'%s'failed.", src.c_str());
